

## Programmes de 1<sup>ère</sup>

Programme de STATISTIQUES n° 1

### NAME=EFFEKT

```
:ClrHome
:Disp "EFFEKT. CUMULES"
:Disp " MOY. ET ME."
:Disp "   ***"
:Disp " (APRES AVOIR"
:Disp " QUITTE, TAPEZ"
:Disp "STAT PUIS EDIT)"
:Pause
:ClrHome
:ClrList (L1,L2,L3,L4,L5,L6)
:0->A
:0->N
:Disp "EFFEKTIF CLASSES"
:Disp " (-1=FIN)"
:Fix 0
:While A<>-1
:N+1->N
:Output(4,11,"N")
:Output(4,12,N)
:Input "= ",A
:If A<>-1
:A->L1(N)
:End
:ClrHome
:Disp "BORNES CLASSES"
:For(I,1,N)
:Output(4,11,"B")
:Output(4,12,I)
:B->C
:Input "= ",B
:B->L6(I)
:If I>1
:Then
:(B+C)/2->X
:X->L2(I-1)
:End
:End
:dim L1->J
:L1(1)->L3(1)
:For(I,2,J)
:L3(I-1)->K
:K+L1(I)->L3(I)
:End
:J->dim L4
:L1(J)->L4(J)
:For(I,1,J-1)
```

```
:L4(J-I+1)->K
:K+L1(J-I)->L4(J-I)
:End
:L1*L2->L5
:sum L5/sumL1->M
:Input "NB DECIMALES :",D
:Fix D
:ClrHome
:Disp "SOMME NI=",sum L1
:Disp "SOMME XINI",sum L5
:Disp "MOYENNE=",M
:Pause
:sum L1/2->N
:1->I
:While L3(I)<N
:I+1->I
:End
:(N-L3(I-1))(L6(I+1)-L6(I))/(L3(I)-L3(I-1))+L6(I)->W
:ClrHome
:Disp "1/2 EFFEKTIF=",N
:Disp "MEDIANE=",W
:Disp ""
:Disp "CF TABLEAUX"
:Float
:ClrList (L6)
```

Programme de résolution de l'équation :

$$ax^2 + bx + c = 0$$

**NAME=DEGRE2**

```
:ClrHome
:Prompt A,B,C
:B^2-4AC->D
:Disp "DELTA=",D
:If D>=0
:Then
:Input "NB DECIMALES :",N
:Fix N
:ClrHome
:End
:If D=0
:Then
:Disp "1 SOL. DOUBLE"
:Disp "X1=X2=", -B/(2A)
:Else
:If D>0
:Then
:Disp "2 SOLUTIONS"
:-B-√D->N
:-B+√D->M
:Disp "X1=",N/(2A)
:Disp "X2=",M/(2A)
:Else
:Disp "PAS DE SOLUTION"
:End
:End
:Float
```

Programme de STATISTIQUES n°2

**NAME=FREQ**

```
:ClrHome
:Disp " FREQUENCES"
:Disp " FCC ET FCD"
:Disp " ***"
:Disp " (APRES AVOIR"
:Disp " QUITTE, TAPEZ"
:Disp "STAT PUIS EDIT)"
:Pause
:ClrHome
:ClrList (L1,L2,L3,L4,L5,L6)
:1->A
:0->N
:Disp " EFFECTIFS"
:Disp " (-1=FIN)"
```

```
:Fix 0
:While A<>-1
:N+1->N
:Output(4,11,"N")
:Output(4,12,N)
:Input "= ",A
:If A<>-1
:A->L1(N)
:End
:sum L1->S
:For(I,1,N-1)
:L1(I)*100/S->F
:F->L2(I)
:End
:dim L1->J
:L2(1)->L3(1)
:For(I,2,J)
:L3(I-1)->K
:K+L2(I)->L3(I)
:End
:J->dim L4
:L2(J)->L4(J)
:For(I,1,J-1)
:L4(J-I+1)->K
:K+L2(J-I)->L4(J-I)
:End
:ClrHome
:Input "NB DECIMALES :",D
:Fix D
:Disp ""
:Disp "CF TABLEAUX"
```

Programme donnant l'équation de la tangente en 1 point d'une courbe :

**NAME=TANGENTE**

```
:ClrHome
:Disp "EQUAT TAN A C(F)"
:Disp "(EQUAT F DS Y1)"
:Prompt X
:nDeriv(Y1,X,X)->A
:Y1-AX->B
:Disp "Y=AX+B"
:Fix 2
:Disp "A",A,"B",B
:Float
```

## Programme de STATISTIQUES n°3

### NAME=MOYET

```

:ClrHome
:Disp " MOYENNE"
:Disp " ECART-TYPE"
:Disp " ***"
:Disp " (APRES AVOIR"
:Disp " QUITTE, TAPEZ"
:Disp "STAT PUIS EDIT)"
:Pause
:ClrHome
:ClrList (L1,L2,L3,L4,L5,L6)
:0->A
:0->N
:Disp " EFFECTIFS :"
:Disp " (-1=FIN)
:Fix 0
:While A<>-1
:N+1->N
:Output(4,11,"N")
:Output(4,12,N)
:Input "=",A
:If A<>-1
:A->L1(N)
:End
:ClrHome
:Menu("VARIABLE","CONTINUE",1,"D
ISCONTINUE",2)
:ClrHome
:Lbl 1
:Disp "BORNES CLASSES :"
:For(I,1,N)
:Output(4,11,"B")
:Output(4,12,I)
:B->C
:Input "=",B
:If I>1
:Then
:(B+C)/2->X
:X->L2(I-1)
:End
:End
:Goto 3
:Lbl 2
:Disp "CLASSES :"
:For(I,1,N-1)
:Output(4,11,"X")
:Output(4,12,I)
:Input "=",X
:X->L2(I)

```

```

:End
:Lbl 3
:L1*L2->L3
:sum L3/sum L1->M
:ClrHome
:Input "NB DECIMALES :",D
:Fix D
:L2-M->L4
:L4^2->L5
:L5*L1->L6
:ClrHome
:Disp "SOMME NI",sum L1
:Disp "SOMME XINI",sum L3
:Disp "SOMME NI(XI-X)^2",sum L6
:Pause
:ClrHome
:Disp "MOYENNE : ",M
: $\sqrt{\text{sum L6/sum L1}}$ ->E
:Disp "ECART TYPE : ",E
:Disp ""
:Disp " CF TABLEAUX"

```

Programme calculant la somme des termes d'une suite arithmétique :

### NAME=SUITEA

```

:ClrHome
:Disp "SUITE ARITHM"
:Input "1ER TERME U1=",U
:Input "RAISON R=",R
:Input "TERME NO=",N
:Disp  $U+(N-1)*R$ 
:Disp "SOMME"
:Input "NB TERMES=",N
:Disp  $N*(U+U+(N-1)*R)/2$ 

```

Programme calculant la somme des termes d'une suite géométrique :

### NAME=SUITEG

```

:ClrHome
:Disp "SUITE GEOMET"
:Input "1ER TERME U1=",U
:Input "RAISON Q=",Q
:Input "TERME NO=",N
:Disp  $U*Q^{(N-1)}$ 
:Disp "SOMME"
:Input "NB TERMES=",N
:Disp  $U*(Q^N-1)/(Q-1)$ 

```

Programme de Résolution d'un système de 2 équations à 2 inconnues :

**NAME=SYST2**

```
:ClrHome
:Disp "SYSTEMES 2EQ 1D"
:Fix 0
:Disp ""
:Disp " COEF. LIGNE 1"
:Prompt A,B,C
:Disp " COEF. LIGNE 2"
:Prompt D,E,F
:ClrHome
:AE-BD->W
:AF-CD->V
:CE-BF->U
:If W=0
:Then
:If U=0
:Then
:Disp "SYST INDETERMINE"
:Else
:Disp "SYST IMPOSSIBLE"
:End
:Else
:ClrHome
:Input "NB DECIMALES :",N
:Fix N
:Disp ""
:Disp "X=",U/W
:Disp "Y=",V/W
:End
:Float
```

Programme de Résolution d'un système de n équations à n inconnues :

**NAME=SYSTEME**

```
:ClrHome
:Disp "SYSTEMES"
:Input "NB EQUAT(->30)=",N
:{N,N}->dim [A]
:dim [A]->dim [C]
:{N,1}->dim [B]
:For(I,1,N)
:ClrHome
:Fix 0
:Disp " COEF. LIGNE"
:Output(1,1,N+1)
:Output(1,15,I)
```

```
:For(J,1,N)
:Input X
:X->[A](I,J)
:X->[C](I,J)
:End
:Input X
:X->[B](I,1)
:X->[C](I,1)
:End
:ClrHome
:Input "NB DECIMALES :",D
:Fix D
:ClrHome
:If abs det [A]>0
:Then
:Disp [A]^-1[B]
:If N<3
:Disp "OU ----"
:If N>3
:Pause
:Disp [A]^(-1)[B]>Frac
:Else
:If abs det [C]>0
:Then
:Disp "SYST IMPOSSIBLE"
:Else
:Disp "SYST INDETERMINE"
:End
:Float
```

Programme calculant l'image en 1 point et le nombre dérivé en 1 point d'une courbe :

**NAME=VALFDF**

```
:ClrHome
:Disp "VALEUR : F ET F'"
:Disp "(EQUAT F DS Y1)"
:Prompt X
:Disp "F(X)="
:Disp Y1
:Disp "F'(X)="
:Disp nDeriv(Y1,X,X)
```

Programme calculant le produit scalaire  $\vec{AB} \cdot \vec{AC}$ , l'angle  $(\vec{AB}, \vec{AC})$  et l'aire du triangle ABC et la longueur du 3<sup>ème</sup> côté BC :

("Résolution" d'un triangle ...)

**NAME=VECTPS**

```

:ClrHome
:Disp ""
:Disp "  VECTEURS"
:Disp "PRODUIT SCALAIRE"
:Disp "  ***"
:Pause
:ClrHome
:Disp "COORD PT ORIG :)"
:Prompt X,Y
:Disp "COORD 2EME PT :)"
:Prompt Z,T
:Disp "COORD 3EME PT :)"
:Prompt U,V
:ClrHome
:Input "NB DECIMALES :",D
:ClrHome
:Disp "COORD VECT 1 :",Z-X,T-Y
:Disp "COORD VECT 2 :",U-X,V-Y
:Pause
:ClrHome
:Fix D
: $\sqrt{(Z-X)^2+(T-Y)^2}$ ->N
:Disp "NORME VECT 1 :",N
: $\sqrt{(U-X)^2+(V-Y)^2}$ ->M
:Disp "NORME VECT 2 :",M
:(Z-X)*(U-X)+(T-Y)*(V-Y)->S
:Disp "P. SCALAIRE :",S
:Pause
:ClrHome
:N*M->P
:S/P->Q
:Degree
:cos-1 Q->A
:Disp "ANGLE (EN DEG)",A
:N*M*sin A*0.5->T
:Disp "AIRE TRIANGLE :",T
: $\sqrt{N^2+M^2-2*M*N*\cos A}$ ->B
:Disp "LONG 3EME COTE :",B
:Float

```